

NAG Fortran Library Routine Document

G05EEF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05EEF sets up the reference vector R for a negative binomial distribution of the number of successes before n failures, where each trial has probability of success p .

2 Specification

```
SUBROUTINE G05EEF(N, P, R, NR, IFAIL)
  INTEGER          N, NR, IFAIL
  real            P, R(NR)
```

3 Description

G05EEF sets up a reference vector for use in G05EYF. Together these routines produce random numbers from the observation defined by:

$$P(I = i) = \frac{(n + i - 1)!}{i!(n - 1)!} p^i (1 - p)^n \quad \text{if } i = 0, 1, \dots,$$

$$P(I = i) = 0 \quad \text{otherwise.}$$

The reference array is generated by a recurrence relation if $np < 50$; otherwise Stirling's approximation is used.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

5 Parameters

- | | | |
|----|---|---------------|
| 1: | N – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the number of failures, n , of the distribution. | |
| | <i>Constraint:</i> $N \geq 0$. | |
| 2: | P – real | <i>Input</i> |
| | <i>On entry:</i> the probability of success, p , of the distribution. | |
| | <i>Constraint:</i> $0.0 \leq P \leq 1.0$. | |
| 3: | R(NR) – real array | <i>Output</i> |
| | <i>On exit:</i> the reference vector. | |
| 4: | NR – INTEGER | <i>Input</i> |
| | <i>On entry:</i> the dimension of the array R as declared in the (sub)program from which G05EEF is called. | |
| | <i>Suggested value:</i> $NR = 20 + (20\sqrt{N \times P} + 30 \times P)/(1 - P)$ approximately (for optimum efficiency in G05EYF). | |

Constraint:

$$\begin{aligned} \text{NR} > & \text{int} \left(\frac{N \times P + 7.15 \sqrt{N \times P} + 20.15 \times P}{1 - P} + 8.5 \right) \\ & - \max \left(0, \text{int} \left(\frac{N \times P - 7.15 \sqrt{N \times P}}{1 - P} \right) \right) + 4. \end{aligned}$$

5: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, –1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value –1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value –1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, $N < 0$.

IFAIL = 2

On entry, $P < 0.0$,
or $P > 1.0$.

IFAIL = 3

On entry, NR is too small (see Section 5).

7 Accuracy

Not applicable.

8 Further Comments

The time taken by the routine increases with NR.

9 Example

The example program sets up a reference vector for a negative binomial distribution with $n = 50$ and $p = 0.5$; it then prints the first five pseudo-random numbers generated by G05EYF, after initialisation by G05CBF.

The generator mechanism used is selected by an initial call to G05ZAF.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G05EEF Example Program Text
*      Mark 20 Revised. NAG Copyright 2001.
*      .. Parameters ..
```

```

      INTEGER          N
      real             P
      INTEGER          NR
      PARAMETER        (N=50,P=0.5e0,NR=250)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*    .. Local Scalars ..
      INTEGER          I, IFAIL, IX
*    .. Local Arrays ..
      real             R(NR)
*    .. External Functions ..
      INTEGER          G05EYF
      EXTERNAL          G05EYF
*    .. External Subroutines ..
      EXTERNAL          G05CBF, G05EEF, G05ZAF
*    .. Executable Statements ..
      CALL G05ZAF('O')
      WRITE (NOUT,*) 'G05EEF Example Program Results'
      WRITE (NOUT,*)
      CALL G05CBF(0)
      IFAIL = 0
*
      CALL G05EEF(N,P,R,NR,IFAIL)
*
      DO 20 I = 1, 5
         IX = G05EYF(R,NR)
         WRITE (NOUT,99999) IX
20    CONTINUE
      STOP
*
99999  FORMAT (1X,I5)
      END

```

9.2 Program Data

None.

9.3 Program Results

G05EEF Example Program Results

```

58
42
46
42
62

```
